
Rich export SPARQL queries: Credential
Note that these are CONSTRUCT queries designed to create a small Jena model for export as a whole after a series of queries has been run.  The 
PERSON_URI variable is substituted by VIVO at runtime.

credential.sparql

PREFIX core: <http://vivoweb.org/ontology/core#>

CONSTRUCT {
     ?credential ?property ?object .
} WHERE {
     PERSON_URI core:relatedBy ?issuedCredential . 
     ?issuedCredential a core:IssuedCredential . 
     ?issuedCredential core:relates ?credential .
     ?credential a core:Credential .
     ?credential ?property ?object .
}

credentialGoverningAuthority.sparql

PREFIX core: <http://vivoweb.org/ontology/core#>

CONSTRUCT {
     ?organization ?property ?object .
} WHERE {
     PERSON_URI core:relatedBy ?issuedCredential . 
     ?issuedCredential a core:IssuedCredential . 
     ?issuedCredential core:relates ?credential .
     ?credential a core:Credential .
     ?credential core:hasGoverningAuthority ?organization .
     ?organization ?property ?object .
} 

elibibleForCredential.sparql

PREFIX core: <http://vivoweb.org/ontology/core#>

CONSTRUCT {
     ?credential ?property ?object .
} WHERE {
     PERSON_URI core:eligibleFor ?credential . 
     ?credential ?property ?object .
}

issuedCredential.sparql

PREFIX core: <http://vivoweb.org/ontology/core#>

CONSTRUCT {
     ?issuedCredential ?property ?object .
} WHERE {
     PERSON_URI core:relatedBy ?issuedCredential . 
     ?issuedCredential a core:IssuedCredential . 
     ?issuedCredential ?property ?object .
}



issuedCredentialExpirationDate.sparql

PREFIX core: <http://vivoweb.org/ontology/core#>

CONSTRUCT {
     ?date ?property ?object .
     ?precision ?property2 ?object2 .
} WHERE {
     PERSON_URI core:relatedBy ?issuedCredential . 
     ?issuedCredential a core:IssuedCredential . 
     ?issuedCredential core:expirationDate ?date .
     ?date ?property ?object .
     ?date core:dateTimePrecision ?precision .
     ?precision ?property2 ?object2 .
}

issuedCredentialIssueDate.sparql

PREFIX core: <http://vivoweb.org/ontology/core#>

CONSTRUCT {
     ?date ?property ?object .
     ?precision ?property2 ?object2 .
} WHERE {
     PERSON_URI core:relatedBy ?issuedCredential . 
     ?issuedCredential a core:IssuedCredential . 
     ?issuedCredential core:dateIssued ?date .
     ?date ?property ?object .
     ?date core:dateTimePrecision ?precision .
     ?precision ?property2 ?object2 .
}

issuedCredentialSubjectArea.sparql

PREFIX core: <http://vivoweb.org/ontology/core#>

CONSTRUCT {
     ?subjectArea ?property ?object .
} WHERE {
     PERSON_URI core:relatedBy ?issuedCredential . 
     ?issuedCredential a core:IssuedCredential . 
     ?issuedCredential core:hasSubjectArea ?subjectArea .
     ?subjectArea ?property ?object .
}

